
Ｊによる花のグラフィックス－２

－朝顔－

西川　利男

０．はじめに

　Ｊによる花のグラフィックス－２として、今真っ盛りの朝顔をとりあげる。残念な

がら、我が家の朝顔は種まきの時期が遅かったのと、手入れがおそまつなので、まだ

ひ弱なつるがやっとからまった程度で花は咲いていない。

　戸川先生の「花のＣＧ」*)で、先の「チューリップ」ではただ２次元の図形に色を塗

っただけであったのが、今度の「朝顔」では花弁を３次元グラフィックスで描いてい

る。したがって私としては３次元グラフィックスの応用として非常に興味のあるテー

マとなった。 　

　　*)戸川隼人著「花のＣＧ」、サイエンス社(1988)

１．朝顔の図形とグラフィックス

　朝顔の花のロート形３次元図形は円筒座標(z, r, θ)で表すのが便利である。しか

し、図形の位置座標の計算処理のためには、３次元直交座標(z, x, y)に変換し、これ

を実際のコンピュータの画面表示では２次元の透視図として描画する。

　具体的には、次のような式になる。

0R と 0Z とはそれぞれ花弁の開口部と高さのパラメータで朝顔の絵に合わせて決めら

れる。

　まずこのようにして、朝顔の花弁の基本の３次元図形が出来上がる。

　つぎに、この図を元に自然な配置になるよう２種類の回転を行なう。

まず、この図を正面（X-Z面）に据えた上で、Y-Z面上で、X軸の周りに角度αだけ手

前に倒す。

　つづいて、X-Z面上でY軸の周りに角度βだけ半時計方向に回す。

　元の図形の座標値(x, y, z)はこの２回の回転による次のような変換行列により新し

い座標値(X, Y, Z)に変換される。

なお今の場合は図形の回転なので、座標軸の回転と間違えないように区別すること。

　図形を回転するための変換行列

・y-z座標面上でx軸の周りに角αの回転

・x-z座標面上でy軸の周りに角βの回転

 上

の

つ

の回転を順次行なうと、次のようになる。

　ここで、朝顔の花弁の曲面図形をどうやって表示するかが問題となる。戸川先生の

有限要素法方式に従い、つまり曲面上の３点を結ぶ小三角形であらわした。

　さらに、裏に隠れた部分のいわゆる陰線処理は、後ろの部分を先に、前の部分を後

から描くことで処理した。

　戸川先生の BASICプログラムは参考にはなったが、Ｊはまったく異なるプログラミ

ングスタイルのため、グラフィック処理はもちろんのこと、値の計算処理もすべて、

新たに別のプログラム作ることとなった。詳細は最後のリスティングを見られたい。

２．朝顔グラフィックスの実際　

　以下に見られるように、Ｊの gl2ウィンドウズにより朝顔グラフィックスを作った。

・ボタン[Go]はまず一輪の朝顔を初期位置、大きさで表示する。

・ボタン[Enter]では、エディット・ボックスが開いて、

　　サイズ、傾き角α、βが自由に入力変更できる。

・ここでボタン[Go]を押すと変更された朝顔が表示される。

・あるいは、グラフィック画面の上、任意の位置でマウスの左ボタンを押すと、

　その位置で朝顔が表示される。

・ボタン[Reset]は初期位置、大きさにリセットされる。

３．朝顔グラフィックスの１ステップ実行

　ボタン[Step]では、途中経過をみるため１ステップずつ実行する。

４．朝顔の回転パラメータα、βの効果の表示

　ボタン[Gallery]では、傾き角α、βの値を変えて比較表示する。

Ｊグラフィックス処理のための配列計算コーディング

　適用例: Ｊのグラフィックス gl2命令gllinesなどの引数として、配列X, Y, Zか

ら点ごとの配列(X0, Y0, Z0), (X1, Y1, Z1),..を必要とする場合が多い。

　配列を Ravel(,)し、Stitch(,.)することで得ることができる。

   X

0 1 2

3 4 5

   Y

10 11 12

13 14 15

   Z

100 101 102

103 104 105

   XYZ =: (,X),.(,Y),.(,Z)

   XYZ

0 10 100

1 11 101

2 12 102

3 13 103

4 14 104

5 15 105

   XY =: }:"(1) XYZ

   XY

0 10

1 11

2 12

3 13

4 14

5 15

   XZ =: }."(1) XYZ

   XZ

10 100

11 101

12 102

13 103

14 104

15 105

   YZ =: ({. , {:)"(1) XYZ

   YZ

0 100

1 101

2 102

3 103

4 104

5 105

朝顔グラフィックス－Ｊプログラム・リスト

NB. asagaoN.ijs　

NB. 2009/7/13, 7/15

NB.   rotation revised by T. N.

require 'trig'

MM =: 10 NB. Z-direction

NN =: 20 NB. XY-circular perfiphery

PI =: 3.1416

ZMIN =: 0

ZMAX =: 180

POW =: 5

CONST =: (ZMAX-ZMIN)%MM^POW

Z0 =: ZMAX-CONST*((i.>:MM)^POW)

T =: 2*PI*(i.>:NN)%NN

RHO =: 1 + 0.07*(|sin (2.5*T))^0.2

RCT =: RHO*cos T

RST =: RHO*sin T

RZERO =: 70

R =: RZERO*(1.03-(i.>:MM)%MM)

XL =: R */ RCT

YL =: R */ RST

ZL =: ($XL)$(({:$XL)#Z0)

NB. make flower

NB. makeflower ALPH0, BETA0 => XX, ZZ

makeflower =: 3 : 0

'ALPHA BETA' =. y.

LX =: - cos BETA

MX =: (sin BETA) * (sin ALPHA)

NX =: - (sin BETA) * (cos ALPHA)

LY =: 0

MY =: cos ALPHA

NY =: sin ALPHA

LZ =: - sin BETA

MZ =: - (cos BETA) * (sin ALPHA)

NZ =: (cos BETA) * (cos ALPHA)

XX =: (LX*XL) + (MX*YL) + (NX*ZL)

XX =: XX,"(1 0) {."(1) XX

XX =: |. XX

YY =: (LY*XL) + (MY*YL) + (NY*ZL)

ZZ =: (LZ*XL) + (MZ*YL) + (NZ*ZL)

ZZ =: ZZ,"(1 0) {."(1) ZZ

ZZ =: |. ZZ

)

require 'gl2'

ASAGAO=: 0 : 0

pc asagao;

menupop "File";

menu new "&New" "" "" "";

menu open "&Open" "" "" "";

menusep ;

menu exit "&Exit" "" "" "";

menupopz;

xywh 211 67 34 12;cc step button;cn "Step";

xywh 211 165 34 12;cc cancel button;cn "Exit";

xywh 10 9 190 173;cc asag isigraph;

xywh 212 14 34 11;cc Go button;

xywh 211 32 34 11;cc Enter button;

xywh 211 144 34 11;cc Clear button;

xywh 211 50 34 11;cc Gallery button;

xywh 210 120 34 11;cc Reset button;

xywh 209 102 34 11;cc Test button;

xywh 210 84 34 11;cc View button;

pas 6 6;pcenter;

rem form end;

)

run=: asagao_run 

asagao_run=: 3 : 0

wd ASAGAO

NB. initialize form here

wd 'pshow;'

asagao_Reset_button ''

)

asagao_close=: 3 : 0

wd'pclose'

)

asagao_cancel_button=: 3 : 0

asagao_close''

- 8 -



)

asagao_Clear_button=: 3 : 0

glclear ''

)

NB. e.g. display_asagao 0 2

display_asagao=: 3 : 0

:

'I J' =. y.

SZZ =. x.

makeflower ALPHA0, BETA0

XA =. (<I, J){XX

XB =. (<I, J+1){XX

XC =. (<(I+1), J){XX

XD =. (<(I+1), (J+1)){XX

ZA =. (<I, J){ZZ

ZB =. (<I, J+1){ZZ

ZC =. (<(I+1), J){ZZ

ZD =. (<(I+1), (J+1)){ZZ

'XA XB XC XD' =. XF + SZZ* XA, XB, XC, XD

'ZA ZB ZC ZD' =. ZF + SZZ* ZA, ZB, ZC, ZD 

VP =. ((XB-XA)*(ZC-ZA))-((XC-XA)*(ZB-ZA))

if. VP > 0

  do.

    glrgb COLA

    glbrush '' 

    glpolygon  XA, ZA, XB, ZB, XC, ZC, XA, ZA

    glpolygon  XD, ZD, XB, ZB, XC, ZC, XD, ZD

  else.

    glrgb COLB

    glbrush ''

    glpolygon  XA, ZA, XB, ZB, XC, ZC, XA, ZA

    glpolygon  XD, ZD, XB, ZB, XC, ZC, XD, ZD

    DP =. % >:i.3

    DQ =. 1 - DP

    DXA =. (DP*XA) + (DQ*XB)

    DXB =. (DP*XC) + (DQ*XD)

    DZA =. (DP*ZA) + (DQ*ZB)

    DZB =. (DP*ZC) + (DQ*ZD)

    XZA =. DXA,.DZA

    XZB =. DXB,.DZB

    glrgb COLB

    gllines L:0 <"(1) XZA,"(1) XZB

  end. 

glshow ''

)

N=: 0

asagao_step_button=: 3 : 0

NN1 =. >: NN

'I J' =. (NN1, NN1)#: N

SZ display_asagao I, J

gltextxy 100, 100

gltext 'XX(',(":I),', ',(":J),') = ', (":(<I, J){XX)

gltextxy 100, 50

gltext 'ZZ(',(":I),', ',(":J),') = ', (":(<I, J){ZZ)

glshow ''

N =: N + 1

)

asagao_Go_button=: 3 : 0

K =. 0

NN1 =. >: NN

while. K < (MM*NN1)

  do.

    'I J' =. (NN1, NN1)#: K

    SZ display_asagao I, J

    K =. K + 1

  end.

)

asagao_Enter_button=: 3 : 0

wd 'pc editbox;'

wd 'xywh 2 11 90 8;cc s0 static;cn "Size, Alpha, Beta?";'

wd 'xywh 100 10 100 10;cc e0 edit ws_border;'

wd 'setfocus e0;'

wd 'pas 8 8;pcenter;pshow;wait;'

wd 'pclose;'

DA =. wd 'q;'

'SZ AL BE' =: ". , >((<'e0') = 0{"(1) DA) # (i.#DA) { 1{"(1) DA

ALPHA0 =: PI*AL%7

BETA0 =: PI%BE

)

asagao_asag_mbldown=: 3 : 0

d=. ". sysdata

x=.(0{d) * 1000 % (2{d)

y=.(1{d) * 1000 % (3{d)

gltextalign TA_BOTTOM

- 10 -



gltext (":<.x), ', ', (":<.y)

'XF ZF' =: x, y

SZ =. 2

COLA =: 255, 125, 125

COLB =: 255, 0, 125

asagao_Go_button ''

)

asagao_Gallery_button=: 3 : 0

COLA =: 255, 200, 200 [COLB =: 255, 0, 125

asagao_Go_button '' [XF =: 400 [ZF =: 700 [SZ =: 1.5 [ALPHA0 =: PI [BETA0 =: 

PI

asagao_Go_button '' [XF =: 400 [ZF =: 500 [SZ =: 1.5 [ALPHA0 =: 5r6*PI [BETA0 

=: PI

asagao_Go_button '' [XF =: 400 [ZF =: 300 [SZ =: 1.5 [ALPHA0 =: 4r6*PI [BETA0 

=: PI

asagao_Go_button '' [XF =: 400 [ZF =: 150 [SZ =: 1.5 [ALPHA0 =: 3r6*PI [BETA0 

=: PI

COLA =: 255, 200, 200 [ COLB =: 255, 0, 255

asagao_Go_button '' [XF =: 250 [ZF =: 500 [SZ =: 1.5 [ALPHA0 =: 5r6*PI [BETA0 

=: 7r6*PI

asagao_Go_button '' [XF =: 550 [ZF =: 500 [SZ =: 1.5 [ALPHA0 =: 5r6*PI [BETA0 

=: 5r6*PI

asagao_Go_button '' [XF =: 700 [ZF =: 500 [SZ =: 1.5 [ALPHA0 =: 5r6*PI [BETA0 

=: 4r6*PI

)

asagao_Reset_button=: 3 : 0

COLA =: 255, 200, 200

COLB =: 255, 0, 125

XF =: 700

ZF =: 50

SZ =: 4

ALPHA0 =: 5r6*PI

BETA0 =: 7r6*PI

glclear ''

)

- 11 -


